CorfuのREADME


- corfu.elのバージョン1.7のREADME.orgの日本語訳です

- 元の文書: https://github.com/minad/corfu/blob/main/README.org

- ライセンス: パッケージのライセンスと同じ、すなわちGPLv3.0ライセンスです

- - -

<a href="https://www.gnu.org/software/emacs/"><img alt="GNU Emacs" src="https://github.com/minad/corfu/blob/screenshots/emacs.svg?raw=true"/></a>

<a href="https://elpa.gnu.org/packages/corfu.html"><img alt="GNU ELPA" src="https://elpa.gnu.org/packages/corfu.svg"/></a>

<a href="https://elpa.gnu.org/devel/corfu.html"><img alt="GNU-devel ELPA" src="https://elpa.gnu.org/devel/corfu.svg"/></a>

<a href="https://melpa.org/#/corfu"><img alt="MELPA" src="https://melpa.org/packages/corfu-badge.svg"/></a>

<a href="https://stable.melpa.org/#/corfu"><img alt="MELPA Stable" src="https://stable.melpa.org/packages/corfu-badge.svg"/></a>

- - -

Table of Contents
=================

* [機能](#機能)
* [インストール](#インストール)
* [キーバインディング](#キーバインディング)
* [コンフィグ](#コンフィグ)
   * [自動補完](#自動補完)
   * [バッファーローカルな補完スタイルやCorfuのみの補完スタイル](#バッファーローカルな補完スタイルやcorfuのみの補完スタイル)
   * [ミニバッファーでの補完](#ミニバッファーでの補完)
   * [EshellやShellでの補完](#eshellやshellでの補完)
   * [Orderlessの補完](#orderlessの補完)
   * [TABのみの補完](#tabのみの補完)
   * [TAB-and-Go補完](#tab-and-go補完)
   * [補完のミニバッファーへの転送](#補完のミニバッファーへの転送)
* [拡張](#拡張)
* [補足的なパッケージ](#補足的なパッケージ)
* [他の選択肢](#他の選択肢)
* [Corfuのデバッグ](#corfuのデバッグ)
* [貢献について](#貢献について)

<!-- Created by https://github.com/ekalinin/github-markdown-toc -->

- - -
<!-- Corfu enhances in-buffer completion with a small completion popup. The current candidates are shown in a popup below or above the point, and can be selected by moving up and down. Corfu is the minimalistic in-buffer completion counterpart of the [Vertico](https://github.com/minad/vertico) minibuffer UI. -->

Corfuは小さな補完ポップアップでバッファー内補完(in-buffer completion)を強化するパッケージだ。カレント候補はポイント位置の下か上にポップアップ表示、上下に移動して選択することができる。CorfuはミニバッファーUIである[Vertico](https://github.com/minad/vertico)と対をなす、ミニマリズム指向のバッファー内補完である。

<!-- Corfu is a small package, which relies on the Emacs completion facilities and concentrates on providing a polished completion UI. In-buffer completion UIs in Emacs can hook into `completion-in-region`, which implements the interaction with the user. Completions at point are either provided by commands like `dabbrev-completion` or by pluggable backends (`completion-at-point-functions`, Capfs) and are then passed to `completion-in-region`. Many programming, text and shell major modes implement a Capf. Corfu does not include its own completion backends. The Emacs built-in Capfs and the Capfs provided by third-party programming language packages are often sufficient. Additional Capfs and completion utilities are provided by the separate [Cape](https://github.com/minad/cape) package. -->

Corfuは補完はEmacsの補完機能に任せて、洗練された補完UIの提供に専念した小さなパッケージだ。Emacsではバッファー内補完UIは`completion-in-region`(ユーザーとの相互作用を実装している)にフックを仕掛けることができる。ポイント位置での補完は`dabbrev-completion`のようなコマンドや、取り外し可能なバックエンド(`completion-at-point-functions`: 以下*Capfs*と略)が提供して、その後に`completion-in-region`に渡されるのだ。Capfsはプログラミング用、テキスト用、シェル用といったメジャーモードの多くで実装されているCorfu自体には補完バックエンドは含まれていない。EmacsビルトインのCapfsと、サードパーティ製のプログラミング言語パッケージが提供するCapfsで間に合うことが多いのだ。追加のCapfsおよび補完ユーティリティは[Cape](https://github.com/minad/cape)パッケージで別途提供している。

<!-- **NOTE**: Corfu relies on child frames to show the popup. Emacs 31 supports child frames also for terminal Emacs. On older Emacs versions, you can use the [corfu-terminal](https://codeberg.org/akib/emacs-corfu-terminal) package. -->


**注意**: Corfuのポップアップの表示は子フレーム(child frame)に依存している。Emacs 31では端末モードのEmacsでも子フレームがサポートされる筈だ。それより前のバージョンのEmacsでも[corfu-terminal](https://codeberg.org/akib/emacs-corfu-terminal)パッケージを使えるだろう。

<img src="https://github.com/minad/corfu/blob/screenshots/light.png?raw=true">

<img src="https://github.com/minad/corfu/blob/screenshots/dark.png?raw=true">

<img src="https://github.com/minad/corfu/blob/screenshots/popupinfo-light.png?raw=true">

<img src="https://github.com/minad/corfu/blob/screenshots/popupinfo-dark.png?raw=true">

# 機能

<!-- -   Timer-based auto-completions (*off* by default). -->
<!-- -   Popup display with scrollbar indicator and arrow key navigation. -->
<!-- -   The popup can be summoned explicitly by pressing `TAB` at any time. -->
<!-- -   The current candidate is inserted with `TAB` and selected with `RET`. -->
<!-- -   Sorting by prefix, string length and alphabetically, optionally by history. -->
<!-- -   The selected candidate is previewed (configurable via `corfu-preview-current`). -->
<!-- -   The selected candidate is automatically committed on further input by default. -->
<!--     (configurable via `corfu-preview-current`). -->
<!-- -   Supports the [Orderless](https://github.com/oantolin/orderless) completion style. The filter string can contain -->
<!--     arbitrary characters, after inserting a space via `M-SPC` (configurable via -->
<!--     `corfu-quit-at-boundary` and `corfu-separator`). -->
<!-- -   Lazy candidate highlighting for performance. -->
<!-- -   Support for candidate annotations (`annotation-function`, `affixation-function`). -->
<!-- -   Deprecated candidates are displayed as crossed out. -->
<!-- -   Icons are provided by external packages via margin formatter functions. -->
<!-- -   Rich set of extensions: Quick keys, Index keys, Sorting by history, Candidate -->
<!--     documentation in echo area, popup or separate buffer. -->

-   タイマーに基づいた自動補完(デフォルトでは *オフ* )

-   インジケーターつきスクロールバーと矢印キーでのナビゲーションができるポップアップ表示

-   `TAB`押下によりいつでも明示的にポップアップを召喚可能

-   カレント候補の挿入は`TAB`、選択は`RET`

-   プレフィックス、文字列長、アルファベット順にソート可能(通常はヒストリー順)

-   選択された候補のプレビュー(`corfu-preview-current`で設定可)

-   デフォルトでは候補選択によりそれ以上の入力は自動省略(`corfu-preview-current`で設定可)

-   補完スタイル[Orderless](https://github.com/oantolin/orderless)をサポート。 `M-SPC`(`corfu-quit-at-boundary`と`corfu-separator`で設定可)の挿入後は、フィルター文字列に任意の文字を含められる。

-   性能に配慮した補完の遅延ハイライト

-   補完への注釈(annotation)のサポート(`annotation-function`と`affixation-function`)

-   鮮度が失われるにつれ候補を抹消

-   marginフォーマット関数経由のサードパーティパッケージ提供のアイコンをサポート

-   インデックスキー、ヒストリー順のソート、エコーエリア、ポップアップ、別バッファーでの候補ドキュメント参照

# インストール

<!-- Corfu is available from [GNU ELPA](https://elpa.gnu.org/packages/corfu.html). You can install it directly via `M-x package-install RET corfu RET`. After installation, activate the global minor mode with `M-x global-corfu-mode RET`. For completion press `M-TAB` (or `TAB`) within a buffer. Auto completion is disabled by default for safety and unobtrusiveness. -->

Corfuは[GNU ELPA](https://elpa.gnu.org/packages/corfu.html)から入手できる。`M-x package-install RET corfu RET`で直接インストールすることも可能だ。インストール後に`M-x global-corfu-mode RET`でグローバルマイナーモードをアクティブにすればよい。バッファーで`M-TAB`(や`TAB`)を翁すれば補完できるだろう。安全と慎みの観点より自動補完はデフォルトでは無効になっている。

# キーバインディング

<!-- Corfu uses a transient keymap `corfu-map` which is active while the popup is shown. The keymap defines the following remappings of fundamental commands and bindings: -->

Corfuは一時的なキーマップ(keymap)として`corfu-map`を使用する。このキーマップはポップアップが表示されている間はアクティブになる。以下の基本的なコマンドやバインディングのリマップ(remap: 再マップ)は、このキーマップで定義されているのだ:

<!-- <table border="2" cellspacing="0" cellpadding="6" rules="groups" frame="hsides"> -->


<!-- <colgroup> -->
<!-- <col  class="org-left" /> -->

<!-- <col  class="org-left" /> -->
<!-- </colgroup> -->
<!-- <thead> -->
<!-- <tr> -->
<!-- <th scope="col" class="org-left">Binding/Remapping</th> -->
<!-- <th scope="col" class="org-left">Corfu command</th> -->
<!-- </tr> -->
<!-- </thead> -->

<!-- <tbody> -->
<!-- <tr> -->
<!-- <td class="org-left"><code>move-beginning-of-line</code></td> -->
<!-- <td class="org-left"><code>corfu-prompt-beginning</code></td> -->
<!-- </tr> -->


<!-- <tr> -->
<!-- <td class="org-left"><code>move-end-of-line</code></td> -->
<!-- <td class="org-left"><code>corfu-prompt-end</code></td> -->
<!-- </tr> -->


<!-- <tr> -->
<!-- <td class="org-left"><code>beginning-of-buffer</code></td> -->
<!-- <td class="org-left"><code>corfu-first</code></td> -->
<!-- </tr> -->


<!-- <tr> -->
<!-- <td class="org-left"><code>end-of-buffer</code></td> -->
<!-- <td class="org-left"><code>corfu-last</code></td> -->
<!-- </tr> -->


<!-- <tr> -->
<!-- <td class="org-left"><code>scroll-down-command</code></td> -->
<!-- <td class="org-left"><code>corfu-scroll-down</code></td> -->
<!-- </tr> -->


<!-- <tr> -->
<!-- <td class="org-left"><code>scroll-up-command</code></td> -->
<!-- <td class="org-left"><code>corfu-scroll-up</code></td> -->
<!-- </tr> -->


<!-- <tr> -->
<!-- <td class="org-left"><code>next-line</code>, <code>down</code>, <code>M-n</code></td> -->
<!-- <td class="org-left"><code>corfu-next</code></td> -->
<!-- </tr> -->


<!-- <tr> -->
<!-- <td class="org-left"><code>previous-line</code>, <code>up</code>, <code>M-p</code></td> -->
<!-- <td class="org-left"><code>corfu-previous</code></td> -->
<!-- </tr> -->


<!-- <tr> -->
<!-- <td class="org-left"><code>completion-at-point</code>, <code>TAB</code></td> -->
<!-- <td class="org-left"><code>corfu-complete</code></td> -->
<!-- </tr> -->


<!-- <tr> -->
<!-- <td class="org-left"><code>M-TAB</code></td> -->
<!-- <td class="org-left"><code>corfu-expand</code></td> -->
<!-- </tr> -->


<!-- <tr> -->
<!-- <td class="org-left"><code>RET</code></td> -->
<!-- <td class="org-left"><code>corfu-insert</code></td> -->
<!-- </tr> -->


<!-- <tr> -->
<!-- <td class="org-left"><code>M-g</code></td> -->
<!-- <td class="org-left"><code>corfu-info-location</code></td> -->
<!-- </tr> -->


<!-- <tr> -->
<!-- <td class="org-left"><code>M-h</code></td> -->
<!-- <td class="org-left"><code>corfu-info-documentation</code></td> -->
<!-- </tr> -->


<!-- <tr> -->
<!-- <td class="org-left"><code>M-SPC</code></td> -->
<!-- <td class="org-left"><code>corfu-insert-separator</code></td> -->
<!-- </tr> -->


<!-- <tr> -->
<!-- <td class="org-left"><code>C-g</code></td> -->
<!-- <td class="org-left"><code>corfu-quit</code></td> -->
<!-- </tr> -->


<!-- <tr> -->
<!-- <td class="org-left"><code>keyboard-escape-quit</code></td> -->
<!-- <td class="org-left"><code>corfu-reset</code></td> -->
<!-- </tr> -->
<!-- </tbody> -->
<!-- </table> -->


<table border="2" cellspacing="0" cellpadding="6" rules="groups" frame="hsides">


<colgroup>
<col  class="org-left" />

<col  class="org-left" />
</colgroup>
<thead>
<tr>
<th scope="col" class="org-left">バインディング/リマップ</th>
<th scope="col" class="org-left">Corfuのコマンド</th>
</tr>
</thead>

<tbody>
<tr>
<td class="org-left"><code>move-beginning-of-line</code></td>
<td class="org-left"><code>corfu-prompt-beginning</code></td>
</tr>


<tr>
<td class="org-left"><code>move-end-of-line</code></td>
<td class="org-left"><code>corfu-prompt-end</code></td>
</tr>


<tr>
<td class="org-left"><code>beginning-of-buffer</code></td>
<td class="org-left"><code>corfu-first</code></td>
</tr>


<tr>
<td class="org-left"><code>end-of-buffer</code></td>
<td class="org-left"><code>corfu-last</code></td>
</tr>


<tr>
<td class="org-left"><code>scroll-down-command</code></td>
<td class="org-left"><code>corfu-scroll-down</code></td>
</tr>


<tr>
<td class="org-left"><code>scroll-up-command</code></td>
<td class="org-left"><code>corfu-scroll-up</code></td>
</tr>


<tr>
<td class="org-left"><code>next-line</code>、<code>down</code>、<code>M-n</code></td>
<td class="org-left"><code>corfu-next</code></td>
</tr>


<tr>
<td class="org-left"><code>previous-line</code>、<code>up</code>、<code>M-p</code></td>
<td class="org-left"><code>corfu-previous</code></td>
</tr>


<tr>
<td class="org-left"><code>completion-at-point</code>、<code>TAB</code></td>
<td class="org-left"><code>corfu-complete</code></td>
</tr>


<tr>
<td class="org-left"><code>M-TAB</code></td>
<td class="org-left"><code>corfu-expand</code></td>
</tr>


<tr>
<td class="org-left"><code>RET</code></td>
<td class="org-left"><code>corfu-insert</code></td>
</tr>


<tr>
<td class="org-left"><code>M-g</code></td>
<td class="org-left"><code>corfu-info-location</code></td>
</tr>


<tr>
<td class="org-left"><code>M-h</code></td>
<td class="org-left"><code>corfu-info-documentation</code></td>
</tr>


<tr>
<td class="org-left"><code>M-SPC</code></td>
<td class="org-left"><code>corfu-insert-separator</code></td>
</tr>


<tr>
<td class="org-left"><code>C-g</code></td>
<td class="org-left"><code>corfu-quit</code></td>
</tr>


<tr>
<td class="org-left"><code>keyboard-escape-quit</code></td>
<td class="org-left"><code>corfu-reset</code></td>
</tr>
</tbody>
</table>

# コンフィグ

<!-- In order to configure Corfu and other packages in your init.el, you may want to use `use-package`. Corfu is flexibly customizable via `corfu-*` customization variables, such that you can adapt it precisely to your requirements. However in order to quickly try out the Corfu completion package, it should be sufficient to activate `global-corfu-mode`. You can experiment with manual completion for example in an Elisp buffer or in an Eshell or Shell buffer. -->

あなたは`use-package`を使って、"init.el"でCorfuと他のパッケージを設定したいと思うかもしれない。カスタマイズ用の変数`corfu-*`を通じて、あなたの要件に正確に適合するようにCorfuを柔軟にカスタマイズできる筈だ。しかしCorfuの補完パッケージを手早く試すだけなら、`global-corfu-mode`をアクティブにすれば十分だろう。たとえばElispバッファーやEshell、あるいはシェルバッファーで手動で補完を経験できる。

<!-- Auto completion is disabled by default in Corfu. Note that completion can be vulnerable to arbitrary code execution in untrusted files. In particular the `elisp-completion-at-point` completion function performs macro expansion and code evaluation. Auto completion can be enabled by setting `corfu-auto` to t locally or globally before enabling the local `corfu-mode` or the `global-corfu-mode`. -->

Corfuでは自動リンクはデフォルトでは無効になっている。補完がもつ信用できないファイルの任意のコードを実行できる能力は、脆弱性となり得ることに注意して欲しい。特にマクロ展開とコードの評価を行う補完関数`elisp-completion-at-point`には注意が必要だ。ローカルでの`corfu-mode`、あるいはグローバルに`global-corfu-mode`を有効にする前に`corfu-auto`に`t`をセットすれば、自動補完が有効になる。以下はコンフィグの例:

```lisp
    (use-package corfu
      ;; オプションのカスタマイズ
      ;; :custom
      ;; (corfu-cycle t)                ;; `corfu-next/previous'による巡回を有効にする
      ;; (corfu-quit-at-boundary nil)   ;; 補完境界で終了しない
      ;; (corfu-quit-no-match nil)      ;; マッチがなくても終了しない
      ;; (corfu-preview-current nil)    ;; 候補のプレビューを無効にする
      ;; (corfu-preselect 'prompt)      ;; プロンプトでの事前選択
      ;; (corfu-on-exact-match nil)     ;; 正確なマッチの取り扱い設定
    
      ;; 特定モードでのみCorfuを有効化(`global-corfu-modes'も参照されたい)
      ;; :hook ((prog-mode . corfu-mode)
      ;;        (shell-mode . corfu-mode)
      ;;        (eshell-mode . corfu-mode))
    
      ;; お勧め: Corfuをグローバルに有効化
      ;; Dabbrevがグローバルに使用可になるのでお勧め
      ;; 特定モードの除外はカスタマイズ変数`global-corfu-modes'を参照のこと
      :init
      (global-corfu-mode))
    
    ;; 更に便利なコンフィグをいくつか...
    (use-package emacs
      :custom
      ;; 候補が少なければTABで巡回
      ;; (completion-cycle-threshold 3)
    
      ;; インデントと補完の両方にTABキーを使える機能を有効化
      ;; `completion-at-point'はM-TABにバインドされていることが多い
      (tab-always-indent 'complete)
    
      ;; Emacs 30以降: Ispellの補完関数を無効化
      ;; かわりに`cape-dict'を試してみて欲しい
      (text-mode-ispell-word-completion nil)
    
      ;; カレントモードには適用されないコマンドはM-xで表示しない
      ;; CorfuコマンドはM-x経由で使用されることはないので非表示になる
      ;; これはCorfuに限らず役に立つセッティングだろう
      (read-extended-command-predicate #'command-completion-default-include-p))
```

`completion-in-region`を基本とするDabbrevはCorfuでも使用可能だ。展開よりも補完の方が好みであれば、簡単にアクセスできるように`dabbrev-completion`と`dabbrev-expand`を交換したいと思うかもしれない。わたしの[Cape](https://github.com/minad/cape)パッケージが提供する、ポイント位置用の補完関数`cape-dabbrev`も試してみて欲しい。

```lisp
    ;; Corfuと一緒にDabbrevを使う!
    (use-package dabbrev
      ;; M-/とC-M-/を交換
      :bind (("M-/" . dabbrev-completion)
             ("C-M-/" . dabbrev-expand))
      :config
      (add-to-list 'dabbrev-ignored-buffer-regexps "\\` ")
      ;; 29.1以前は古い`dabbrev-ignored-buffer-regexps'を使用すること
      (add-to-list 'dabbrev-ignored-buffer-modes 'doc-view-mode)
      (add-to-list 'dabbrev-ignored-buffer-modes 'pdf-view-mode)
      (add-to-list 'dabbrev-ignored-buffer-modes 'tags-table-mode))
```

<!-- If you start to configure Corfu more thoroughly, I recommend to give the -->
<!-- Orderless completion style a try for filtering. Orderless completion offers more -->
<!-- flexible filtering than the default completion styles. Note that Orderless is -->
<!-- not a necessity; Corfu can be used just as well with the default completion -->
<!-- styles. -->

より徹底したCorfu設定を始める場合には、フィルタリングに補完スタイルOrderlessを試すことをお勧めする。Orderlessの補完はデフォルトの補完スタイルい比べてより柔軟なフィルタリングを提供する。Orderlessが必要という訳ではないことに注意。デフォルトの補完スタイルでも、同じようにCorfuは使用できるのだ。

```lisp
    ;; オプションで補完スタイル`orderless'を使用する
    (use-package orderless
      :custom
      ;; (orderless-style-dispatchers '(orderless-affix-dispatch))
      ;; (orderless-component-separator #'orderless-escapable-split-on-space)
      (completion-styles '(orderless basic))
      (completion-category-defaults nil)
      (completion-category-overrides '((file (styles partial-completion)))))
```

<!-- The `basic` completion style is specified as fallback in addition to `orderless` in order to ensure that completion commands which rely on dynamic completion tables, e.g., `completion-table-dynamic` or `completion-table-in-turn`, work correctly. Additionally enable `partial-completion` for file path expansion. `partial-completion` is important for file wildcard support. Multiple files can be opened at once with `find-file` if you enter a wildcard. You may also give the `initials` completion style a try. -->

`orderless`に加えてフォールバックとして補完スタイル`basic`を指定しているが、これは動的補完テーブル(`completion-table-dynamic`や`completion-table-in-turn`)に依存する補完コマンドの、正しい動作を保証するためだ。ファイルパスの展開用に追加で`partial-completion`も有効にしている。ファイルでのワイルドカードをサポートするのであれば`partial-completion`は重要だ。`find-file`でワイルドカードを入力すれば、複数ファイルを一度にオープンできる。補完スタイル`initials`を試してみるのもよいだろう。

<!-- See also the [Corfu Wiki](https://github.com/minad/corfu/wiki) and the [Cape manual](https://github.com/minad/cape) for additional Capf configuration tips. For more general documentation read the chapter about completion in the [Emacs manual](https://www.gnu.org/software/emacs/manual/html_node/emacs/Completion.html). If you want to create your own Capfs, you can find documentation about completion in the [Elisp manual](https://www.gnu.org/software/emacs/manual/html_node/elisp/Completion.html).  -->

Capf用の追加コンフィグのヒントについては[Corfu Wiki](https://github.com/minad/corfu/wiki)や[Cape manual](https://github.com/minad/cape)にもあるので参考にして欲しい。より一般的な文書については、[Emacs manual](https://www.gnu.org/software/emacs/manual/html_node/emacs/Completion.html)の補完のチャプターを読むとよいだろう。独自のCapfsを作成したい場合には、[Elisp manual](https://www.gnu.org/software/emacs/manual/html_node/elisp/Completion.html)で補完に関する文書が見つかる筈だ。

## 自動補完

<!-- Auto completion is disabled by default for safety and unobtrusiveness. Note that completion can be vulnerable to arbitrary code execution. Auto completion can be enabled by setting `corfu-auto` to t. Only enable auto completion locally in trusted buffers or globally if you edit trusted files only. -->

安全性と慎みの観点から、自動補完はデフォルトでは無効になっている。任意のコードを実行できるという能力から、補完が脆弱性となり得ることに注意して欲しい。`corfu-auto`に`t`をセットすれば自動補完は有効にできる。自動補完は信頼できるバッファーでのみローカルに有効にするか、グローバルに有効にする場合には信頼できるファイルだけを編集しているときに限定する必要がある。

<!-- You may want to configure Corfu to quit completion eagerly, such that the completion popup stays out of your way when it appeared unexpectedly. -->

意図せず補完がポップアップして作業の邪魔になるような場合には、補完をさっさと閉じるようにCorfuを設定したいだろう。

```lisp
    ;; 自動補完を有効化するとともに補完終了を設定
    (setq corfu-auto t
          corfu-quit-no-match 'separator) ;; またはt
```

<!-- I suggest to experiment with the various settings and key bindings to find a configuration which works for you. There is no one perfect configuration which fits all. Some people like auto completion, some like manual completion, some want to cycle with TAB and some with the arrow keys. -->


あなたにとって満足できるセッティングが見つかるまで、さまざまなセッティングやキーバインディングを体験してみることをお勧めする。すべてを満足するような完璧な設定など存在しない。自動補完を好む人もいれば手動補完を好む人、TABで巡回したい人や矢印キーで巡回したい人もいるのだから。

<!-- In case you like auto completion settings, where the completion popup appears immediately, better use a cheap completion style like `basic`, which performs prefix filtering. See the next section about setting Corfu-only completion styles. In this case Corfu completion should still be fast in buffers with efficient completion backends. You can try the following settings in an Elisp buffer or the Emacs scratch buffer. Note that such settings can slow down Emacs due to the high load on the Lisp runtime and garbage collector. -->

補完が即座にポップアップする自動補完セッティングがお好みならば、`basic`(プレフィックスによりフィルタリングを行う)のような低コストの補完スタイルを使用する方がよいだろう。Corfuに限定した補完スタイルのセッティングについては次のセクションで説明しよう。このような場合には、効率的な補完バックエンドと組み合わせれば、バッファー内補完のCorfuの方が依然として高速な筈だ。ElispバッファーやEmacsのscratchバッファーで以下のセッティングを試してみることができる。この種のセッティングは高負荷なLispの実行時ロードとガベージコレクションにより、Emacs全体の速度が低下する恐れがあることに注意。

```lisp
    (setq corfu-auto        t
          corfu-auto-delay  0  ;; 小さすぎ - 非推奨!
          corfu-auto-prefix 0) ;; 小さすぎ - 非推奨!
    
    (add-hook 'corfu-mode-hook
              (lambda ()
                ;; Corfuだけのセッティング
                (setq-local completion-styles '(basic)
                            completion-category-overrides nil
                            completion-category-defaults nil)))
```

## バッファーローカルな補完スタイルやCorfuのみの補完スタイル

<!-- Sometimes it makes sense to use separate completion style settings for minibuffer completion and in-buffer Corfu completion. For example inside the minibuffer you may prefer advanced Orderless completion, while for Corfu, faster prefix completion is needed or literal-only completion is sufficient. -->

ミニバッファー用の補完とバッファー内のCorfu補完用に、補完スタイルを個別に使い分けられた方がよい場合がある。たとえば高速なプレフィックス補完が必要だったりリテラルの補完だけで十分な場合にはCorfu、ミニバッファーの中では高度なOrderless補完を使いたい場合だ。

<!-- This matters in particular if you use aggressive auto completion settings, where the completion popup appears immediately. Then a cheap completion style like `basic` should be used, which performs prefix filtering only. -->

これは補完が即座にポップアップ表示される積極的(aggressive)なセッティングでは特に影響があるだろう。そのような場合にはプレフィックスによるフィルタリングしか行わない、`basic`のような低コストの補完スタイルを使用するべきだ。

<!-- Such Corfu-only configurations are possible by setting the `completion-styles` variables buffer-locally, as follows: -->

このようなCorfuだけのコンフィグは、バッファーローカルに変数`completion-styles`を設定することで実現できるだろう。

```lisp
    (orderless-define-completion-style orderless-literal-only
      (orderless-style-dispatchers nil)
      (orderless-matching-styles '(orderless-literal)))
    
    (add-hook 'corfu-mode-hook
              (lambda ()
                (setq-local completion-styles '(orderless-literal-only basic)
                            completion-category-overrides nil
                            completion-category-defaults nil)))
```

<!-- If you want to combine fast prefix filtering and Orderless filtering you can still do that by defining a custom Orderless completion style via `orderless-define-completion-style`. We use a custom style dispatcher, which enables efficient prefix filtering for input shorter than 4 characters. -->

高速なプレフィックスによるフィルタリングとOrderlessのフィルタリングを組み合わせたい場合には、まだ`orderless-define-completion-style`を通じてOrderlessの補完スタイルをカスタム定義するという手が残されている。4文字よりも短い入力にたいして効率的なプレフィックスフィルタリングを有効にする、カスタムのスタイルディスパッチャを使うことにしよう。

```lisp
    (defun orderless-fast-dispatch (word index total)
      (and (= index 0) (= total 1) (length< word 4)
           (cons 'orderless-literal-prefix word)))
    
    (orderless-define-completion-style orderless-fast
      (orderless-style-dispatchers '(orderless-fast-dispatch))
      (orderless-matching-styles '(orderless-literal orderless-regexp)))
    
    (setq corfu-auto        t
          corfu-auto-delay  0  ;; 小さすぎ - 非推奨!
          corfu-auto-prefix 0) ;; 小さすぎ - 非推奨!
    
    (add-hook 'corfu-mode-hook
              (lambda ()
                (setq-local completion-styles '(orderless-fast basic)
                            completion-category-overrides nil
                            completion-category-defaults nil)))
```

## ミニバッファーでの補完

<!-- Corfu can be used for completion in the minibuffer, since it relies on child frames to display the candidates. The Corfu popup floats on top of the Emacs frame and can be shown even if it doesn't fit inside the minibuffer. -->

Corfuの候補表示は子フレーム頼りである。それ故に逆にミニバッファーの補完にCorfuを使用することができることになる。CorfuのポップアップはEmacsの最上位フレームにフローティング表示されるので、ミニバッファー内部に収まらなくても表示が可能なのだ。


<!-- `global-corfu-mode` activates `corfu-mode` in the minibuffer if the variable `global-corfu-minibuffer` is non-nil. In order to avoid interference with specialised minibuffer completion UIs like Vertico or Mct, Corfu is only enabled if the minibuffer sets the variable `completion-at-point-functions` locally. This way minibuffers with completion can be detected, such that minibuffer commands like `M-:` (`eval-expression`) or `M-!` (`shell-command`) are enhanced with Corfu completion. -->

変数`global-corfu-minibuffer`が非`nil`の場合には、`global-corfu-mode`をアクティブにすることでミニバッファーで`corfu-mode`もアクティブになる。VerticoやMctのようにミニバッファーに特化した補完UIと干渉しないように、ミニバッファーで変数`completion-at-point-functions`がローカルにセットされた場合のみCorfuが有効化される仕組みだ。この方法なら補完付きのミニバッファーを検知できるので、`M-:` (`eval-expression`)や`M-!` (`shell-command`)のようなコマンドをCorfuの補完で補強できる筈だ。

<!-- If needed, one can also enable Corfu more generally in all minibuffers, as long as no completion UI is active. In the following example we set `global-corfu-minibuffer` to a predicate function, which checks for Mct and Vertico. Furthermore we ensure that Corfu is not enabled if a password is read from the minibuffer. -->

必要ならより一般的に、アクティブな補完UIがなくなるまでCorfuをすべてのミニバッファーで有効にすることもできる。以下の例では、MctとVerticoをチェックする述語関数として`global-corfu-minibuffer`をセットしている。更にミニバッファーからの読み取ったのがパスワードの場合には、確実にCorfuが無効になるようにしてみよう。

```lisp
    (setq global-corfu-minibuffer
          (lambda ()
            (not (or (bound-and-true-p mct--active)
                     (bound-and-true-p vertico--input)
                     (eq (current-local-map) read-passwd-map)))))
```

## EshellやShellでの補完

<!-- When completing in the Eshell I recommend conservative local settings without auto completion, such that the completion behavior is similar to widely used shells like Bash, Zsh or Fish. -->

Eshellで行う際には自動補完なしにすることをお勧めする。広く使われているBash、Zsh、あるいはFishと同じような、控え目な振る舞いになるようなローカルセッティングにしよう。

```lisp
    (add-hook 'eshell-mode-hook (lambda ()
                                  (setq-local corfu-auto nil)
                                  (corfu-mode)))
```

<!-- When pressing `RET` while the Corfu popup is visible, the `corfu-insert` command will be invoked. This command does inserts the currently selected candidate, but it does not send the prompt input to Eshell or the Comint process. Therefore you often have to press `RET` twice which feels like an unnecessary double confirmation. Fortunately it is easy to improve this by using the command `corfu-send` instead. -->

Corfuのポップアップ表示中に`RET`を押下すると、`corfu-insert`コマンドが呼び出される筈だ。これはカレントで選択された候補を挿入するコマンドだが、EshellやCominitのプロセスにプロンプトの入力を送信しない。したがって2回の押下が必要なことがよくある。この二重確認は不要に思える。幸い`corfu-send`コマンドをかわりに使えば、これは簡単に改善できるのだ。

```lisp
    (keymap-set corfu-map "RET" #'corfu-send)
```

<!-- Shell completion uses the flexible Pcomplete mechanism internally, which allows you to program the completions per shell command. If you want to know more, look into this [blog post](https://www.masteringemacs.org/article/pcomplete-context-sensitive-completion-emacs), which shows how to configure Pcomplete for git commands. Since Emacs 29, Pcomplete offers the `pcomplete-from-help` function which parses the `--help` output of a command and produces completions for command line options. -->

シェル補完が内部的に使用しているのは、柔軟なPcompleteメカニズムである。これはシェルコマンドごとにプログラミング可能な補完だ。より詳細について知りたければ、[blog post](https://www.masteringemacs.org/article/pcomplete-context-sensitive-completion-emacs)を調べて欲しい。この記事ではgitコマンド用のPcomplete設定にも触れている。Emacsの29以降のPcompleteはコマンドにたいして`--help`を渡したときの出力を解析して、コマンドラインオプションに使用できる補完を生成できる。

<!-- Pcomplete has a few bugs on Emacs 28 and older. We can work around the issues with the [Cape](https://github.com/minad/cape) library (Completion at point extensions). Cape provides wrappers which sanitize the Pcomplete function. On Emacs 29 the advices should not be necessary anymore, since most relevant bugs have been fixed. In case you discover any remaining Pcomplete issues, please report them upstream. -->

Emacs 28以前のPcompleteにはバグがいくつか存在する。この問題は[Cape](https://github.com/minad/cape)ライブラリー(ポイント位置補完の拡張)で解決できるだろう。Capeが提供するのはPcomplete関数を消毒するためのラッパーだ。Emacs 29では関係のあるほとんどのバグは修正されたので、adviceはもう不要な筈だ。まだ残っているPcompleteの問題を見つけたら、どうかアップストリームに報告していただきたい。

```lisp
    ;; `pcomplete-completions-at-point'(以降Capfと略)を消毒
    ;; このCapfにはEmacs 28以前では望ましくない副作用があった
    ;; Emacs 29以降ではこれらのadviceは不要になった
    (when (< emacs-major-version 29)
      (advice-add 'pcomplete-completions-at-point :around #'cape-wrap-silent)
      (advice-add 'pcomplete-completions-at-point :around #'cape-wrap-purify))
```

## Orderlessの補完

<!-- [Orderless](https://github.com/oantolin/orderless) is an advanced completion style that supports multi-component search filters separated by a configurable character (space, by default). Normally, entering characters like space which lie outside the completion region boundaries (words, typically) causes Corfu to quit. This behavior is helpful with auto-completion, which may pop-up when not desired, e.g. on entering a new variable name. Just keep typing and Corfu will get out of the way. -->

[Orderless](https://github.com/oantolin/orderless)がサポートしているのは、設定可能な文字(デフォルトではスペース)で区切られた、マルチコンポーネントの検索フィルターという補完スタイルだ。通常は補完リージョンの境界外にあるスペースのような文字を入力することで、Corfuは終了する。これは自動補完では役に立つ挙動だろう。たとえば新たな変数名をタイプしたときは意図せずポップアップするかもしれない。それでもタイプを続行すれば、Corfuのポップアップは消えるだろうから。

<!-- But orderless search terms can contain arbitrary characters; they are also interpreted as regular expressions. To use orderless, set `corfu-separator` (a space, by default) to the primary character of your orderless component separator. -->

しかしOrderlessの検索には特定の任意な文字を含めることができ、それらは正規表現としても解釈されるのだ。Orderlessを使用するには、`corfu-separator`にOrderlessの主要なコンポーネント区切り(デフォルトはスペース)をセットする。

<!-- Then, when a new orderless component is desired, use `M-SPC` (`corfu-insert-separator`) to enter the first component separator in the input, and arbitrary orderless search terms and new separators can be entered thereafter.  -->

そうすれば新たなOrderlessコンポーネントが必要になったら、`M-SPC` (`corfu-insert-separator`)を使って1つ目のコンポーネントの区切りを入力する。その後にOrderlessの任意の検索用語を続けて、また新たな区切りを入力するといった具合に使用するのだ。

<!-- To treat the entire input as Orderless input, you can set the customization option `corfu-quit-at-boundary` to nil. This disables the predicate which checks if the current completion boundary has been left. In contrast, if you always want to quit at the boundary, set `corfu-quit-at-boundary` to t. By default `corfu-quit-at-boundary` is set to `separator` which quits at completion boundaries as long as no separator has been inserted with `corfu-insert-separator`. -->

入力全体をOrderlessの入力として扱わせたい場合にはカスタマイズ用のオプション`corfu-quit-at-boundary`に`nil`をセットすればよい。これはカレントの補完境界が残っていないかどうかチェックする述語を無効にするオプションだ。反対に常に境界で終了させたい場合には、`corfu-quit-at-boundary`に`t`をセットすればよいだろう。デフォルトでは`corfu-quit-at-boundary`は`separator`にセットされている。これは`corfu-insert-separator`で区切りが入力されないかぎりは、補完境界で終了するオプションだ。

<!-- Finally, there exists the user option `corfu-quit-no-match` which is set to `separator` by default. With this setting Corfu stays alive as soon as you start advanced filtering with a `corfu-separator` even if there are no matches, for example due to a typo. As long as no separator character has been inserted with `corfu-insert-separator`, Corfu will still quit if there are no matches. This ensures that the Corfu popup goes away quickly if completion is not possible. -->

最後に`corfu-quit-no-match`(デフォルトは`separator`)というユーザーオプションがある。このセッティングなら、`corfu-separator`で高度な検索を開始したCorfuは、たとえばタイプの間違え等のせいでマッチがなくてもそのまま生き残る筈だ。しかし`corfu-insert-separator`で区切り文字を挿入した場合には、マッチがなければ終了する挙動は変わらない。これにより補完が不可能なら、Corfuが速やかにポップアップを消すことが保証されるのだ。

<!-- In the following we show two configurations, one which works best with auto completion and one which may work better with manual completion if you prefer to always use `SPC` to separate the Orderless components. -->

以下に2つのコンフィグを紹介しよう。1つ目は自動補完においてベストなコンフィグ、もう1つはOrderless補完の区切りに常に`SPC`を使いたい場合には上手く機能しそうなコンフィグだ。

```lisp
    ;; 自動補完の例
    (use-package corfu
      :custom
      (corfu-auto t)          ;; 自動補完を有効化
      ;; (corfu-separator ?_) ;; スペース以外の区切りを使うならセット
      :bind
      ;; S-SPCのように別のキーにバインドしてもよい
      ;; (:map corfu-map ("M-SPC" . corfu-insert-separator))
      :init
      (global-corfu-mode))
    
    ;; 手動補完の例
    (use-package corfu
      :custom
      ;; (corfu-separator ?_) ;; Orderlessの区切りにスペース以外を使うならセット
      :bind
      ;; SPCで区切る設定
      (:map corfu-map ("SPC" . corfu-insert-separator))
      :init
      (global-corfu-mode))
```

## TABのみの補完

<!-- By default, Corfu steals both the `RET` and `TAB` keys, when the Corfu popup is open. This can feel intrusive, in particular in combination with auto completion. `RET` may accidentally commit an automatically selected candidate, while you actually wanted to start a new line. As an alternative we can unbind the `RET` key completely from `corfu-map` or reserve the `RET` key only in shell modes using a menu-item filter. -->

デフォルトではCorfuがポップアップをオープン中には、`RET`と`TAB`の両方のキーをCorfuが横取りする。これは特に自動補完と組み合わせた場合には、差し出がましく感じるかもしれない。実際には改行したいのに、自動選択された候補を`RET`で間違えてコミットしてしまう恐れもあるだろう。

`corfu-map`から`RET`を完全にバインド解除したり、シェルモードでのみ複数アイテム用のフィルターで用いるように`RET`を予約するという別の手もある

```lisp
    ;; TABだけの設定
    (use-package corfu
      :custom
      (corfu-auto t)               ;; 自動補完を有効化
      (corfu-preselect 'directory) ;; ディレクトリー以外は最初の候補を選択
    
      :init
      (global-corfu-mode)
    
      :config
      ;; RETキーの開放で少し慎ましく
      ;; Option 1: RETのバインドを完全に解除
      ;; (keymap-unset corfu-map "RET")
      ;; Option 2: シェルモードのみRETを使用する
      (keymap-set corfu-map "RET" `( menu-item "" nil :filter
                                     ,(lambda (&optional _)
                                        (and (derived-mode-p 'eshell-mode 'comint-mode)
                                             #'corfu-send)))))
```

## TAB-and-Go補完

<!-- You may be interested in configuring Corfu in TAB-and-Go style. Pressing TAB moves to the next candidate and further input will then commit the selection. Note that further input will not expand snippets or templates, which may not be desired but which leads overall to a more predictable behavior. In order to force snippet expansion, confirm a candidate explicitly with `RET`. -->

TAB-and-GoスタイルでCorfuを設定することに興味があるだろうか。TABの押下で次の候補へ移動、更に押下すると選択が確定するのだ。スニペットの展開を強制するには、`RET`で明示的に候補を選ぶようにしよう。

```lisp
    (use-package corfu
      ;; TAB-and-Goの設定画面
      :custom
      (corfu-cycle t)           ;; `corfu-next/previous'での巡回を有効化
      (corfu-preselect 'prompt) ;; プロンプトは常に自動選択
    
      ;; TAB(デフォルトは`corfu-complete'だった)は巡回に使用
      :bind
      (:map corfu-map
            ("TAB" . corfu-next)
            ([tab] . corfu-next)
            ("S-TAB" . corfu-previous)
            ([backtab] . corfu-previous))
    
      :init
      (global-corfu-mode))
```

## 補完のミニバッファーへの転送

<!-- Sometimes it is useful to transfer the Corfu completion session to the minibuffer, since the minibuffer offers richer interaction features. In particular, [Embark](https://github.com/oantolin/embark) is available in the minibuffer, such that you can act on the candidates or export/collect the candidates to a separate buffer. We could add Corfu support to Embark in the future, such that export or collect is possible directly from Corfu. Nevertheless, the ability to transfer the Corfu completion to the minibuffer is even more powerful, since further completion is possible. -->

より豪華な対話機能を提供するミニバッファーに、Corfuの補完セッションを転送すると役に立つ場合もあるのではないだろうか。特にミニバッファーでは[Embark](https://github.com/oantolin/embark)が利用できるのだ。候補にアクションを適用したり、候補を別のバッファーにエクスポートしたり収集することができるだろう。直接エクスポートや収集ができるEmbarkのサポートは、近い将来Corfuに追加されるだろう。とはいえCorfuの補完をミニバッファーに転送できれば更に強力になる筈だ。ミニバッファーでのさらなる補完が可能になるからだ。

<!-- The command `corfu-move-to-minibuffer` is defined here in terms of `consult-completion-in-region`, which uses the minibuffer completion UI via `completing-read`. -->

ここでコマンド`corfu-move-to-minibuffer`は、(`completing-read`を通じてミニバッファー補完を使用する)`consult-completion-in-region`の観点で定義してある。

```lisp
    (defun corfu-move-to-minibuffer ()
      (interactive)
      (pcase completion-in-region--data
        (`(,beg ,end ,table ,pred ,extras)
         (let ((completion-extra-properties extras)
               completion-cycle-threshold completion-cycling)
           (consult-completion-in-region beg end table pred)))))
    (keymap-set corfu-map "M-m" #'corfu-move-to-minibuffer)
    (add-to-list 'corfu-continue-commands #'corfu-move-to-minibuffer)
```

# 拡張

<!-- We maintain small extension packages to Corfu in this repository in the subdirectory [extensions/](https://github.com/minad/corfu/tree/main/extensions). The extensions are installed together with Corfu if you pull the package from ELPA. The extensions are inactive by default and can be enabled manually if desired. Furthermore it is possible to install all of the files separately, both `corfu.el` and the `corfu-*.el` extensions. Currently the following extensions come with the Corfu ELPA package: -->

このレポジトリのサブディレクトリー[extensions/](https://github.com/minad/corfu/tree/main/extensions)には、わたしたちが保守している小さな拡張パッケージがある。ELPAからパッケージをインストールしたのなら、Corfuと一緒に拡張もインストールされている筈だ。拡張はデフォルトではアクティブになっていないが、アクティブにしたいなら手動で有効にできる。更に`corfu.el`と`corfu-*.el`両方の拡張ファイルは、すべて個別にインストール可能だ。ELPAのCorfuパッケージには、現在のところ以下の拡張が同梱されている:

<!-- -   [corfu-echo](https://github.com/minad/corfu/blob/main/extensions/corfu-echo.el): `corfu-echo-mode` displays a brief candidate documentation in the echo area. -->
<!-- -   [corfu-history](https://github.com/minad/corfu/blob/main/extensions/corfu-history.el): `corfu-history-mode` remembers selected candidates and sorts the candidates by their history position. -->
<!-- -   [corfu-indexed](https://github.com/minad/corfu/blob/main/extensions/corfu-indexed.el): `corfu-indexed-mode` allows you to select indexed candidates with prefix arguments. -->
<!-- -   [corfu-info](https://github.com/minad/corfu/blob/main/extensions/corfu-info.el): Actions to access the candidate location and documentation. -->
<!-- -   [corfu-popupinfo](https://github.com/minad/corfu/blob/main/extensions/corfu-popupinfo.el): Display candidate documentation or source in a popup next to the candidate menu. -->
<!-- -   [corfu-quick](https://github.com/minad/corfu/blob/main/extensions/corfu-quick.el): Commands to select using Avy-style quick keys. -->

-   [corfu-echo](https://github.com/minad/corfu/blob/main/extensions/corfu-echo.el): `corfu-echo-mode`はエコーエリアに補完の要約ドキュメントを表示する

-   [corfu-history](https://github.com/minad/corfu/blob/main/extensions/corfu-history.el): `corfu-history-mode`は選択した候補を覚えておいて、候補のヒストリー上の位置で候補をソートする

-   [corfu-indexed](https://github.com/minad/corfu/blob/main/extensions/corfu-indexed.el): `corfu-indexed-mode`はインデックス付けした候補をプレフィックスで選ぶようにできる

-   [corfu-info](https://github.com/minad/corfu/blob/main/extensions/corfu-info.el):候補の位置やドキュメントにアクセスするためのアクション

-   [corfu-popupinfo](https://github.com/minad/corfu/blob/main/extensions/corfu-popupinfo.el): 補完メニュー隣りに候補のドキュメントとソースをポップアップ表示

-   [corfu-quick](https://github.com/minad/corfu/blob/main/extensions/corfu-quick.el): Avyスタイルのクイックキーを用いた選択コマンド

<!-- See the Commentary of those files for configuration details. -->

コンフィグの詳細は、これらのファイルのコメントを参照して欲しい。

# 補足的なパッケージ

<!-- Corfu works well together with all packages providing code completion via the `completion-at-point-functions`. Many modes and packages already provide a Capf out of the box. Nevertheless you may want to look into complementary packages to enhance your setup. -->

Corfuは`completion-at-point-functions`を通じたコード補完を提供するすべてのパッケージと良好に動作する筈だ。インストールしただけで使用できるCapfを、すでに多くのモードやパッケージが提供している。とはいえセットアップの改善に、補足的なパッケージについても知っておきたいのではないだろうか。

<!-- -   [corfu-candidate-overlay](https://code.bsdgeek.org/adam/corfu-candidate-overlay): Shows as-you-type auto-suggestion candidate overlay with a visual indication of whether there are many or exactly one candidate available (works only with `corfu-auto` disabled). -->

<!-- -   [corfu-terminal](https://codeberg.org/akib/emacs-corfu-terminal): Child frames are supported by terminal Emacs 31 out of the box. On older Emacs versions, this package provides an overlay-based popup display. -->

<!-- -   [Orderless](https://github.com/oantolin/orderless): Corfu supports completion styles, including the advanced `orderless` completion style, where the filter expressions are separated by spaces or another character (see `corfu-separator`). -->

<!-- -   [Cape](https://github.com/minad/cape): Provides additional Capf backends and `completion-in-region` commands. Among others, the package supplies the file completion backend `cape-file` and the Dabbrev backend `cape-dabbrev`. Cape provides the `cape-company-to-capf` adapter to reuse Company backends in Corfu. -->

<!-- -   [kind-icon](https://github.com/jdtsmith/kind-icon), [nerd-icons-corfu](https://github.com/LuigiPiucco/nerd-icons-corfu): Icons are supported by Corfu via external packages. The nerd-icons-corfu package relies on the Nerd icon font, which is even supported on terminal, while kind-icon uses SVGs from monochromatic icon sets. -->

<!-- -   [Tempel](https://github.com/minad/tempel): Tiny template/snippet package with templates in Lisp syntax, which can be used in conjunction with Corfu. -->

<!-- -   [Vertico](https://github.com/minad/vertico): You may also want to look into my Vertico package. Vertico is the minibuffer completion counterpart of Corfu. -->


-   [corfu-candidate-overlay](https://code.bsdgeek.org/adam/corfu-candidate-overlay): タイプするにしたがい、多くの候補の中に目指す候補が存在するのかを視覚的に示す補完オーバーレイを表示する(`corfu-auto`が無効な場合のみ機能する)。

-   [corfu-terminal](https://codeberg.org/akib/emacs-corfu-terminal): Emacs 31からはデフォルトで端末の子フレームがサポートされる。それより古いバージョンのEmacsにたいしてオーバーレイベースのポップアップ表示を提供するパッケージだ。

-   [Orderless](https://github.com/oantolin/orderless): Corfuがサポートしている補完スタイルには、`orderless`の上級向けな補完スタイルが含まれている(スペースあるいは別の文字でフィルター式を区切れる; `corfu-separator`を参照)

-   [Cape](https://github.com/minad/cape): 追加のCapfバックエンドと`completion-in-region`を提供する。他のパッケージと異なるのはファイル補完バックエンドの`cape-file`、そしてDabbrevバックエンドの`cape-dabbrev`を提供する点だ。更にCapeはCorfuでCompanyのバックエンドを再利用する際のアダプターとなる`cape-company-to-capf`も提供している。

-   [kind-icon](https://github.com/jdtsmith/kind-icon)および[nerd-icons-corfu](https://github.com/LuigiPiucco/nerd-icons-corfu): Corfuでは外部パッケージを通じてアイコンをサポートしている。kind-iconがサポートするのはモノクロのアイコンセットに基づくSVGであることに比べて、nerd-icons-corfuは端末さえもサポートするNerdアイコンを基にしたパッケージだ。

-   [Tempel](https://github.com/minad/tempel): Lisp構文のテンプレートでテンプレート/スニペットを提供する小さなパッケージ。Corfuと組み合わせて使用可能。

-   [Vertico](https://github.com/minad/vertico): わたしのVerticoパッケージにも興味があるかもしれない。VerticoはCorfuと対を成すミニバッファー補完を提供する。

# 他の選択肢

<!-- -   [Company](https://github.com/company-mode/company-mode): Company is a widely used and mature completion package, which implements a similar UI as Corfu. While Corfu relies exclusively on the standard Emacs completion API (Capfs), Company defines its own API for the backends. Company includes its own completion backends, following its own API, which are incompatible with the Emacs completion infrastructure. Company provides an adapter `company-capf` to handle Capfs as a Company backend. As a result of this design, Company is a more complex package than Corfu, three times as large, even without backends. Company by default uses overlays for the popup in contrast to the child frames used by Corfu. Overall both packages work well, but Company integrates less tightly with Emacs. The `completion-styles` support is more limited and the `completion-at-point` command and the `completion-in-region` function do not invoke Company -->.

-   [Company](https://github.com/company-mode/company-mode): Companyは広く使われている成熟した補完パッケージであり、Corfuと似たUIを実装している。Corfuは専らEmacsの標準補完API(Capfs)に頼っているが、Companyはバックエンド用に独自にAPIを定義している。Companyには自身の独自APIにしたがう独自の補完バックエンドがあるがこれはEmacsの補完インストラクチャーとは互換性がないのだ。CompanyはCompanyのバックエンドとしてCapfsを処理するアダプターを提供している。このデザインの結果、CompanyはCorfuに比べてより複雑であり、バックエンドを除外してもおよそ3倍のサイズがある。またCorfuではポップアップに子フレームを用いるが、Companyはデフォルトではポップアップにオーバーレイを使用する点も対照的だろう。いずれのパッケージも全体としては良好に動作するが、Emacsとの統合の度合いという点においてはCompanyが若干劣っている。`completion-styles`のサポートはより制限されており、コマンド`completion-at-point`および関数`completion-in-region`はCompanyを呼び出さないのだ。

<!-- -   [consult-completion-in-region](https://github.com/minad/consult): The Consult package provides the function `consult-completion-in-region` which can be set as `completion-in-region-function` such that it handles `completion-at-point`. The function works by transferring the in-buffer completion to the minibuffer. In the minibuffer, the minibuffer completion UI, for example [Vertico](https://github.com/minad/vertico) takes over. If you prefer to perform all your completions in the minibuffer `consult-completion-in-region` is your best option. -->

-   [consult-completion-in-region](https://github.com/minad/consult): Consultパッケージは`consult-completion-in-region`という関数を提供している。`completion-in-region-function`にこの関数をセットすれば、`completion-at-point`の処理を任せることができる。この関数はバッファー内の補完をミニバッファーに転送することによって機能する。ミニバッファーにおいてはたとえば[Vertico](https://github.com/minad/vertico)がミニバッファー用の補完UIとして使用されるだろう。補完はすべてミニバッファーで行いたければ、`consult-completion-in-region`が最善のオプションだ。

# Corfuのデバッグ

<!-- When you observe an error in the `corfu--post-command` post command hook, you should install an advice to enforce debugging. This allows you to obtain a stack trace in order to narrow down the location of the error. The reason is that post command hooks are automatically disabled (and not debugged) by Emacs. Otherwise Emacs would become unusable, given that the hooks are executed after every command. -->

もしあなたがポストコマンドフック`corfu--post-command`でエラーを発見したら、デバッグを強制するためにadviceをインストールする必要がある。これによりエラー位置を絞り込むためのスタックトレースが入手できる筈だ。これはEmacsによってポストコマンドフックが自動的に無効にされる(そしてデバッグされない)せいだ。さもないとすべてのコマンドの後にこのフックが実行されてしまい、Emacsが使い物にならなくなってしまうだろう。

```lisp
    (setq debug-on-error t)
    
    (defun force-debug (func &rest args)
      (condition-case e
          (apply func args)
        ((debug error) (signal (car e) (cdr e)))))
    
    (advice-add #'corfu--post-command :around #'force-debug)
```

<!-- When Capfs do not yield the expected result you can use `cape-capf-debug` to add debug messages to a Capf. The Capf will then produce a completion log in the messages buffer. -->

思ったようにCapfsの結果が得られない場合には、Capfへのデバッグメッセージ追加に`cape-capf-debug`を付かうことができるだろう。。そうすればCapfが補完ログをmessageバッファーに出力する筈だ

```lisp
    (setq completion-at-point-functions (list (cape-capf-debug #'cape-dict)))
```

<!-- Note that you will sometimes find crashes inside Capfs. Such issues are bugs in the Capfs must be fixed there. They cannot be worked around in Corfu. -->

Capfs内部のクラッシュを目にするかもしれないことに注意。そのような問題は修正しなければならないバグがそのCapfsに存在するということだ。このようなCapfsはCorfuで使うことはできない。

# 貢献について

<!-- Since this package is part of [GNU ELPA](https://elpa.gnu.org/packages/corfu.html) contributions require a copyright assignment to the FSF.  -->

このパッケージは[GNU ELPA](https://elpa.gnu.org/packages/corfu.html)の一部なので、貢献にはFSFへの著作権譲渡が必要ないとなる。